

Pianificazione di Produzione in DEC

L'esempio considerato qui è un problema reale che la Digital Equipment Corporation (DEC) ha dovuto affrontare nell'autunno del 1988 per preparare la pianificazione di produzione per l'inverno.

Nella primavera dello stesso anno, la DEC aveva introdotto una nuova linea di workstation e server con CPU singola: GP-1, GP-2 e GP-3 (server con differenti caratteristiche) e WS-1 e WS-2 (workstation). Le caratteristiche sono riassunte nella tabella sotto.

Sistema	Prezzo	Media dischi fissi	Moduli di RAM
GP-1	60	0.3	4
GP-2	40	1.7	2
GP-3	30	0	2
WS-1	30	1.4	2
WS-2	15	0	1

Il prezzo è in migliaia di dollari, e la “media dischi fissi” è il numero medio di dischi fissi montati su ogni unità: per esempio, tutti i GP-2 hanno almeno un disco fisso, e 7 su 10 ne hanno 2.

Le consegne di questa nuova linea di prodotti erano iniziate durante l'estate ed incrementate lentamente nell'autunno. Per l'inverno, erano state anticipate le seguenti difficoltà.

- (i) La produzione interna di CPU aveva limitato la produzione a 7000 unità per via di problemi di debugging.
- (ii) La fornitura di dischi fissi era stata stimata nella gamma 3000-7000 unità.
- (iii) La fornitura dei moduli di RAM era limitata alla gamma 8000-16000 unità.

La sezione marketing aveva stabilito che la domanda durante l'inverno sarebbe stata: 1800 GP-1, 300 GP-3 e 3800 unità di tutta la classe “server”; e 3200 unità di tutta la classe “workstation”. Inclusi in queste proiezioni c'erano anche 500 ordini di GP-2, 500 di WS-1 e 400 di WS-2 che erano già stati ricevuti ma non ancora spediti ai clienti. Nei precedenti periodi di pianificazione, per far fronte alla limitazione di dischi fissi, la DEC aveva prodotto GP-1, GP-3 e WS-2 senza dischi fissi (anche se 3 clienti su 10 richiedevano GP-1 con il disco fisso) e GP-2 e WS-1 con soltanto un disco fisso. Questa configurazione viene chiamata “produzione vincolata”.

Per l'inverno, la DEC avrebbe dovuto prendere le decisioni seguenti.

- (a) Il piano di produzione per il periodo invernale.

- (b) Era opportuno continuare a costruire in “produzione vincolata” o era meglio cercare di soddisfare le richieste dei clienti?
- (c) Era meglio cercare di concentrare gli sforzi per produrre più dischi fissi o più moduli di memoria?

Prevedendo il caso peggiore sulla produzione di dischi fissi e moduli di RAM, si formuli un modello di PL per prendere le decisioni (a) e (b) sopra. Si usino le tecniche di analisi di sensitività per prendere la decisione (c). I problemi di PL possono essere risolti numericamente mediante AMPL e CPLEX.

[Suggerimento: per prendere la decisione (b), si risolva il problema con due distinti file di dati. La “produzione vincolata” corrisponde a valori di “media dischi fissi” diversi da quelli dati nella tabella sopra.]

Soluzione

FORMULAZIONE

- *Indici:* sia i un indice sull'insieme $\{1, \dots, N\}$ di tipi di computer (nell'istanza descritta qui, $N = 5$). Sia $n \leq N$ un numero tale che i computer $1, \dots, n$ sono server e i computer $n + 1, \dots, N$ sono workstation (nell'istanza descritta qui, $n = 3$).
- *Parametri:*
 - p_i è il prezzo di vendita del computer i (per ogni $i \leq N$);
 - H_i è il numero medio di dischi fissi nel computer i (per ogni $i \leq N$);
 - M_i è il numero di moduli di RAM richiesti nel computer i (per ogni $i \leq N$);
 - c è il numero di CPU disponibili;
 - m è il numero di moduli di RAM disponibili;
 - h è il numero di dischi fissi disponibili;
 - d_i è la domanda minima per il computer i (per ogni $i \leq N$);
 - D_i è la domanda massima per il computer i (per ogni $i \leq N$);
 - S è la domanda massima per i server;
 - W è la domanda massima per le workstation.

- *Variabili:* x_i è la quantità di computer di tipo i da produrre

- *Funzione obiettivo:*

$$\max \sum_{i=1}^N p_i x_i$$

- *Vincoli:*

1. (disp. CPU) $\sum_{i=1}^N x_i \leq c$;
2. (disp. RAM) $\sum_{i=1}^N M_i x_i \leq m$;
3. (disp. dischi fissi) $\sum_{i=1}^N H_i x_i \leq h$;
4. (domanda min/max) per ogni $i \leq N$, $d_i \leq x_i \leq D_i$;
5. (domanda max server) $\sum_{i=1}^n x_i \leq S$;
6. (domanda max workstation) $\sum_{i=n+1}^N x_i \leq W$;
7. (limiti sulle variabili) per ogni $i \leq N$, $x_i \geq 0$.

Si noti innanzitutto che sebbene il problema sia di natura intera (numero di computer da produrre), le quantità sono tali che è possibile approssimare il problema usando variabili continue.

In secondo luogo, le domande minime sono date soltanto per GP-2, WS-1, WS-2 (ordini già accettati): si pongano le altre domande minime a zero. Le domande massime sono date solo per GP-1 e GP-3. Si pongano le altre domande massime a un numero molto grande (per es., il numero c di CPU, dato che la produzione totale di tutti i computer non può superare c) in modo che il vincolo corrispondente sia inattivo.

Terzo: la decisione (b) deve essere presa precedentemente alla decisione (a): prima bisogna decidere se usare la produzione vincolata oppure no, e soltanto dopo si può derivare il modello della pianificazione di produzione. In pratica, bisogna risolvere due problemi distinti: uno con la produzione vincolata e uno senza. La soluzione migliore determinerà la decisione (b) e contemporaneamente, anche il piano di produzione. I due problemi sono istanze diverse (“dati diversi”) dello stesso modello riportato sopra: infatti la produzione vincolata corrisponde a un diverso valore del parametro H_i (media di dischi fissi sul computer i) in cui $H_1 = H_3 = H_5 = 0$, $H_2 = H_4 = 1$. In pratica, vanno preparati due file di dati diversi: uno corrispondente alla produzione vincolata e uno corrispondente alla scelta di privilegiare la preferenza dei clienti. Il problema va poi risolto con entrambi i file di dati per verificare quale alternativa sia migliore.

MODELLO AMPL

```
# problema dec
param computer default 5;
set N := 1..computer;
param n integer, >= 0, <= computer, default 3;
param p{N} >= 0;
param H{N} >= 0;
param M{N} >= 0;
param c >= 0, integer;
param m >= 0, integer;
param h >= 0, integer;
param d{N} >= 0;
param D{N} >= 0;
param S >= 0;
param W >= 0;
var x{N} >= 0;
maximize ricavo: sum{i in N} p[i]*x[i];
subject to cpu: sum{i in N} x[i] <= c;
subject to ram: sum{i in N} M[i]*x[i] <= m;
subject to dischi: sum{i in N} H[i]*x[i] <= h;
subject to dmin{i in N}: x[i] >= d[i];
subject to dmax{i in N}: x[i] <= D[i];
subject to server: sum{i in N : i <= n} x[i] <= S;
subject to workstation: sum{i in N : i >= n} x[i] <= W;
```

DATI AMPL PER PRODUZIONE VINCOLATA

```
# problema dec: dati con produzione vincolata
```

```
param computer := 5;
param n := 3;
param :
  p H M d D :=
  1 60 0.0 4 0 1800
  2 40 1.0 2 500 7000
  3 30 0.0 2 0 300
  4 30 1.0 2 500 7000
  5 15 0.0 1 400 7000 ;
param c := 7000;
param m := 8000;
param h := 3000;
param S := 3800;
param W := 3200;
```

DATI AMPL SENZA PRODUZIONE VINCOLATA

```
# problema dec: dati senza produzione vincolata
param computer := 5;
param n := 3;
param :
  p H M d D :=
  1 60 0.3 4 0 1800
  2 40 1.7 2 500 7000
  3 30 0.0 2 0 300
  4 30 1.4 2 500 7000
  5 15 0.0 1 400 7000 ;
param c := 7000;
param m := 8000;
param h := 3000;
param S := 3800;
param W := 3200;
```

PROGRAMMA AMPL

```
model dec.mod;
data dec_pv.dat;
option solver cplex;
solve;
display ricavo;
display x;
reset;
model dec.mod;
data dec.dat;
solve;
display ricavo;
display x;
```

SOLUZIONE

```
OptiRisk Systems license manager: valid AMPL license found
ILOG CPLEX 8.100, licensed to "politecnico-milano", options: e m b q
CPLEX 8.1.0: optimal solution; objective 145000
2 dual simplex iterations (0 in phase I)
ricavo = 145000
```

```
x [*] :=
1  400
2  2500
3    0
4  500
5  400
;
```

```
ILOG CPLEX 8.100, licensed to "politecnico-milano", options: e m b q
CPLEX 8.1.0: optimal solution; objective 132903.2258
4 dual simplex iterations (0 in phase I)
ricavo = 132903
```

```
x [*] :=
1  354.839
2  1290.32
3   300
4   500
5  2400
;
```

Dai risultati si evince che conviene il modalità di produzione vincolata; il piano di produzione per il periodo invernale è dunque: $x = (400, 2500, 0, 500, 400)$.

Per prendere la decisione (c), dobbiamo verificare la variazione sulla funzione obiettivo di un incremento unitario di h (limite sui dischi fissi) e di m (limite sui moduli di memoria). Chiediamo quindi ad AMPL il valore delle variabili duali corrispondenti ai vincoli di disponibilità dischi fissi e disponibilità RAM:

```
display dischi, ram;
```

Si verifica che la variabile duale del vincolo sui dischi fissi vale 10 all'ottimo del problema con produzione vincolata, mentre la variabile duale del vincolo sulla RAM vale 15. È perciò opportuno concentrare gli sforzi sulla produzione di moduli di memoria RAM.